Electronic Supplementary Material (ESI) for Molecular BioSystems
This journal is © The Royal Society of Chemistry 2011

Appendix S3: Rules and molecule type definitions for example visualizations

This appendix provides rules and molecule type definitions encoded in the BioNetGen language (BNGL) [1] to accompany Figs. 6-9 in the main text. These figures are included in this appendix for easy reference. The interaction arrows in the figures are numbered to correspond to rules. In some cases, multiple ways of characterizing an interaction are presented.



This journal is © The Royal Society of Chemistry 2011

Fig. 6A: Protein synthesis and interaction of a transcription factor with a DNA binding site.

Transcription factor binds DNA:

```
TF(DNA_binding_domain) + DNA(promoter) <->
TF(DNA_binding_domain!1).DNA(promoter!1) k_on_DNA,k_off_DNA
```
(1)

Protein synthesis:

```
TF(DNA_binding_domain!1).DNA(promoter!1) ->
TF(DNA_binding_domain!1).DNA(promoter!1) + protein() k_synth
```
(2a)

Alternatively, a Hill function can be used, in which `k`, `K`, and `n` represent parameters in the function:

```
TF() -> TF() + protein() Hill(k,K,n)
```
(2b)

Fig. 6B: Degradation

Ubiquitinated protein is degraded:

```
protein(K~Ub) -> 0 k_deg
```
(3)

A degradation rule must include a product because BioNetGen requires that at least one product be specified in every rule. In this example, `0` is used, which is the recommended syntax in BioNetGen version 2.1.9.

Fig. 6C: Proteolysis

Caspase-10 binds caspase-3:

```
caspase10(CASC) + caspase3(CASC) <->
caspase10(CASC!1).caspase3(CASC!1)
k_on_cas, k_off_cas
```
(4a)

Caspase-10 cleaves caspase-3, liberating p17 and p12. Here, the reaction is represented as degradation of caspase-3 and synthesis of p17 and p12.

```
caspase10(CASC!1).caspase3(CASC!1) ->
caspase10(CASC) + p17() + p12() kcat_cas
```
(4b)

Alternatively, caspase-3 can be represented as a complex of p17 and p12 and Eqs. 4a and 4b can be replaced with a single rule that has a Michaelis-Menten (`MM`) rate law. The following rule represents

This journal is © The Royal Society of Chemistry 2011

cleavage of the (covalent) bond between p17 and p12:

```
caspase10() + p17(D175!1).p12(S176!1) ->
caspase10() + p17(D175) + p12(S176)                    (4c)
MM(kcat_cas,Km_cas)
```

Note that the parameter `Km_cas` can be defined as `(k_off_cas+kcat_cas)/k_on_cas` for consistency with the rules given above.

The Michaelis-Menten mechanism is represented exactly in Eqs. 4a and 4b and represented approximately, by the `MM` rate law, in Eq. 4c. There are many potential pitfalls associated with using a Michaelis-Menten rate law because of the assumptions involved [2–4], and without knowledge of parameter values, it is not possible to determine whether the Michaelis-Menten rate law provides a good approximation of the kinetics of the underlying reaction mechanism.

Fig. 6D: Allosteric regulation and small-molecule metabolism

PFK-1 binds fructose 2,6-bisphosphate (F(2,6)BP), which is modeled as having a virtual binding site, `bind`.

```
PFK1(allosteric) + F26BP(bind) <->
PFK1(allosteric!1).F26BP(bind!1)                       (5)
k_on_F26BP,k_off_F26BP
```

PFK-1 binds fructose-6-phosphate (F6P). PFK-1 bound to F(2,6)BP has a higher affinity for F6P. Two different rules are used to account for this contextual difference. Rule 6a represents binding of PFK-1 to F6P without allosteric activation. Rule 6b represents binding of activated PFK-1 to F6P. F6P is modeled as having a virtual binding site, `bind`.

```
PFK1(active_site,allosteric) + F6P(bind) <->
PFK1(active_site!1,allosteric).F6P(bind!1)             (6a)
k_on_PFK1,k_off_PFK1
```

```
PFK1(active_site,allosteric!1).F26BP(bind!1) + F6P(bind) <->
PFK1(active_site!2,allosteric!1).F26BP(bind!1).F6P(bind!2)   (6b)
k_on_PFK1_act,k_off_PFK1_act
```

PFK-1 phosphorylates F6P, generating F(1,6)BP. This reaction is represented as degradation of F6P and synthesis of F(1,6)P.

```
PFK1(active_site!1).F6P(bind!1) ->
PFK1(active_site) + F16P kcat_PFK1                     (6c)
```

As an alternative to Eqs. 6a-c, Michaelis-Menten rate laws can be used to characterize phosphorylation of F6P by PFK-1:

```
PFK1(allosteric) + F6P -> PFK1(allosteric) + F16P MM(kcat_PFK1,Km_PFK1)   (6d)
```

This journal is © The Royal Society of Chemistry 2011

Note that `Km_PFK1` can be defined as `(k_off_PFK1+kcat_PFK1)/k_on_PFK1` for consistency with Eqs. 6a and 6c.

```
PFK1(allosteric!1).F26BP(bind!1) + F6P ->
PFK1(allosteric!1).F26BP(bind!1) + F16P                    (6e)
MM(kcat_PFK1,Km_PFK1_act)
```

Note that `Km_PFK1_act` can be defined as `(k_off_PFK1_act+kcat_PFK1_act)/k_on_PFK1_act` for consistency with Eqs. 6b and 6c. As mentioned previously, a Michaelis-Menten rate law is not guaranteed to be a good approximation of kinetics of a Michaelis-Menten reaction mechanism [2–4].

Fig. 6E: Phosphorylation and dephosphorylation of lipids

In Eqs. 7a-9b, the two phospholipid species PIP2 (phosphatidylinositol 4,5-bisphosphate) and PIP3 (phosphatidylinositol (3,4,5)-trisphosphate) are represented with a single molecule type definition, `PtdIns`. This molecule contains three components that represent carbon atoms at positions 3, 4, and 5 in an inositol ring. PIP2 is phosphorylated at positions 4 and 5 only, whereas PIP3 is phosphorylated at positions 3, 4, and 5.

PI3K binds PIP2:

```
PI3K(PI3KC) + PtdIns(3~0,4~P,5~P) <->
PI3K(PI3KC!1).PtdIns(3~0!1,4~P,5~P)                        (7a)
k_on_PI3K,k_off_PI3K
```

PI3K phosphorylates PIP2 at the 3' position of the inositol ring, generating PIP3. This reaction is represented as a change in the phosphorylation state of the carbon at position 3:

```
PI3K(PI3KC!1).PtdIns(3~0!1,4~P,5~P) ->
PI3K(PI3KC) + PtdIns(3~P,4~P,5~P) k_phos_PI3K              (7b)
```

PIP3 binds the PH domain of PDK1:

```
PtdIns(3~P,4~P,5~P) + PDK1(PH) <->
PtdIns(3~P!1,4~P,5~P).PDK1(PH!1)                           (8)
k_on_PDK1,k_off_PDK1
```

PTEN binds PIP3:

```
PTEN(phos) + PtdIns(3~P,4~P,5~P) <->
PTEN(phos!1).PtdIns(3~P!1,4~P,5~P)                         (9a)
k_on_PTEN,k_off_PTEN
```

PTEN dephosphorylates PIP3, generating PIP2:

```
PTEN(phos!1).PtdIns(3~P!1,4~P,5~P) ->
PTEN(phos) + PtdIns(3~0,4~P,5~P) kdephos_PTEN              (9b)
```

This journal is © The Royal Society of Chemistry 2011

Fig. 6F: Dephosphorylation of proteins

SHP-1 binds Lck:

```
SHP1(PTP) + Lck(Y394~P) <->
SHP1(PTP!1).Lck(Y394~P!1)
k_on_SHP1,k_off_SHP1
```
(10a)

SHP-1 dephosphorylates Lck:

```
SHP1(PTP!1).Lck(Y394~P!1) ->
SHP1(PTP) + Lck(Y394~0) kcat_SHP1
```
(10b)

Alternatively, Eqs. 10a and 10b can be replaced with a single rule that has a Michaelis-Menten (`MM`) rate law:

```
SHP1 + Lck(Y394~P ->
SHP1 + Lck(Y394~0)
MM(kcat_SHP1,Km_SHP1)
```
(10c)

Note that the parameter `Km_SHP1` can be defined as `(k_off_SHP1+kcat_SHP1)/k_on_SHP1` for consistency with Eqs. 10a and 10b. As mentioned previously, a Michaelis-Menten rate law is not guaranteed to be a good approximation of the kinetics of a Michaelis-Menten reaction mechanism [2–4].

Fig. 6G: Transport

NF-κB is transported to the nucleus when IκB is not bound to its RHD domain. Two possible ways of representing this process are shown below.

Compartmental BNGL (cBNGL) [5] allows for explicit representation of compartments. Here, `Cyt` represents cytoplasm and `Nuc` represents the nucleus. Inclusion of the `RHD` domain, without bonds to other molecules, represents a contextual constraint; to be transported to the nucleus, NF-κB must not be free (not bound to IκB):

```
@Cyt:NFKB(RHD) -> @Nuc:NFKB(RHD) ktrans
```
(11a)

Alternatively, compartments can be represented as states of a location component, `loc`.

```
NFKB(RHD,loc~Cyt) -> NFKB(RHD,loc~Nuc) ktrans
```
(11b)

Note that `loc` is a virtual component, as it does not represent an actual material component of NF-κB.

This journal is © The Royal Society of Chemistry 2011



Fig. 7A, B, C: Ubiquitination cascade

Equations 12-16 provide a generic representation of a ubiquitination cascade. Classes of enzymes (e.g., E1) are used instead of specific enzyme names. Binding sites (e.g., `E2_bind`) are virtual, in that the names of these components are not intended to identify the actual material components of proteins responsible for direct physical interactions. Such virtual binding sites are useful when the real binding sites are unknown. Figure 7c shows a specific case where a subset of these rules may be applied.

E1 binds E2:

```
E1(E2_bind) + E2(E1_bind) <->
E1(E2_bind!1).E2(E1_bind!1)                    (12)
k_on_E1E2, k_off_E1E2
```

This journal is © The Royal Society of Chemistry 2011

Ubiquitin is transferred from E1 to E2 and ubiquitin is represented as a state:

```
E1(C~Ub,E2_bind!1).E2(C~0,E1_bind!1) ->
E1(C~0,E2_bind) + E2(C~Ub,E1_bind) ktrans_E1E2
```
(13)

E2 binds E3:

```
E2(E3_bind) + E3(E2_bind) <->
E2(E3_bind!1).E3(E2_bind!1)
k_on_E2E3, k_off_E2E3
```
(14)

E3 binds target:

```
E3(target_bind) + target(E3_bind) <->
E3(target_bind!1).target(E3_bind!1)
k_on_E3target,k_off_E3target
```
(15)

Ub is transferred from E2 to target:

```
E2(C~Ub,E3_bind!1).E3(E2_bind!1,target_bind!2).target(K~0,E3_bind!2) ->
E2(C~0,E3_bind!1).E3(E2_bind!1,target_bind) + target(K~Ub,E3_bind)
ktrans_E2target
```
(16)

Fig. 7D: Transfer of a ubiquitin-like protein

In Eqs. 17-20b, the components `Atg10_bind`, `Atg7_bind`, `Atg5_bind` are virtual.

Atg7 binds Atg10:

```
Atg7(Atg10_bind) + Atg10(Atg7_bind) <->
Atg7(Atg10_bind!1).Atg10(Atg7_bind!1)
k_on_Atg7Atg10,k_off_Atg7Atg10
```
(17)

Atg7 catalyzes transfer of Atg12 to Atg10. Here, Atg12 is treated as a state of C507 and C133 in Atg7 and Atg10, respectively:

```
Atg7(Atg10_bind!1,C507~Atg12).Atg10(Atg7_bind!1,C133~0) ->
Atg7(Atg10_bind,C507~0) + Atg10(Atg7_bind,C133~Atg12)
ktrans_Atg7Atg10
```
(18a)

Alternatively, Atg12 can be represented as an individual molecule that interacts with Atg7 and Atg10 via its G196 residue:

```
Atg7(Atg10_bind!1,C507!2).Atg12(G196!2).Atg10(Atg7_bind!1,C133) ->
Atg7(Atg10_bind,C507) + Atg12(G196!2).Atg10(Atg7_bind,C133!2)
ktrans_Atg7Atg10
```
(18b)

Atg10 binds Atg5:

```
Atg10(Atg5_bind) + Atg5(Atg10_bind) <->
Atg10(Atg5_bind!1).Atg5(Atg10_bind!1)
k_on_Atg10Atg5,k_off_Atg10Atg5
```
(19)

This journal is © The Royal Society of Chemistry 2011

Atg10 catalyzes transfer of Atg12 to Atg5. As in Eq. 18a, Atg12 is treated as a state:

```
Atg10(Atg5_bind!1,C133~Atg12).Atg5(Atg10_bind!1,K149~0) ->
Atg10(Atg5_bind,C133~0) + Atg5(Atg10_bind,K149~Atg12)
ktrans_Atg10Atg5
```
(20a)

Alternatively, as in Eq. 18b, Atg12 can be treated as a molecule:

```
Atg10(Atg5_bind!1,C133!2).Atg12(G196!2).Atg5(Atg10_bind!1,K149) ->
Atg10(Atg5_bind,C133) + Atg12(G196!2).Atg5(Atg10_bind,K149!2)
ktrans_Atg10Atg5
```
(20b)

This journal is © The Royal Society of Chemistry 2011



Fig. 8: Ras

In Eqs. 21-26, the guanine nucleotide loading state of HRas is treated as an internal state of a `GTPase` component of HRas. The HRas GTPase domain has three possible internal states: `0` (empty), `GDP` (bound to GDP), and `GTP` (bound to GTP). `GTPase~GTP` and `GTPase~GDP` both represent a protein bound to a guanine nucleotide via a noncovalent interaction. Accordingly, interaction arrows representing noncovalent bonds are drawn between GTP, GDP, and HRas in Fig. 8. It is worth noting that in previous examples and in Appendix S1, internal states were used to represent covalent modifications (e.g., `Y~P` for a phosphorylated tyrosine residue), and were visualized with covalent modification flags. Rules alone do not indicate whether an internal state represents a change arising from covalent or noncovalent bonds. In contrast, an extended contact map distinguishes between covalent and noncovalent bonds through the use of arrows and modification flags.

$p120^{RasGAP}$ binds HRas, which activates the intrinsic GTPase activity of HRas. $p120^{RasGAP}$ has different affinities for GTP-loaded and GDP-loaded HRas:

```
HRas(GTPase~GTP) + RasGAP(GAP) <->
HRas(GTPase~GTP!1).RasGAP(GAP!1)                    (21a)
k_on_GAPGTP,k_off_GAPGTP

HRas(GTPase~GDP) + RasGAP(GAP) <->
HRas(GTPase~GDP!1).RasGAP(GAP!1)                    (21b)
k_on_GAPGDP,k_off_GAPGDP
```

Activated HRas hydrolyzes GTP:

This journal is © The Royal Society of Chemistry 2011

```
HRas(GTPase~GTP!1).RasGAP(GAP!1) ->
HRas(GTPase~GDP!1).RasGAP(GAP!1)                                    (22)
k_GTPGDP
```

HRas binds the REM domain of Sos1, where it acts as an allosteric activator. To specify that HRas must be GDP-loaded or GTP-loaded to bind Sos1, one can use two rules that specify internal states of the `GTPase` component. Here, for simplicity, we assume that HRas binding to Sos1 is independent of whether HRas is bound to GDP or GTP. As a result, the following two rules have the same kinetic parameters:

```
HRas(GTPase~GDP) + Sos1(REM) <->
HRas(GTPase~GDP!1).Sos1(REM!1)                                      (23a)
k_on_RasREM,k_off_RasREM
```

```
HRas(GTPase~GTP) + Sos1(REM) <->
HRas(GTPase~GTP!1).Sos1(REM!1)                                      (23b)
k_on_RasREM,k_off_RasREM
```

Alternatively, one can use the `exclude_reactants` command to modify rule application:

```
HRas(GTPase) + Sos1(REM) <-> HRas(GTPase!1).Sos1(REM!1)                                  (23c)
k_on_RasREM,k_off_RasREM exclude_reactants(1,HRas(GTPase~0))
```

where `1` corresponds to the index of the first reactant (i.e., HRas) and is followed by a pattern that is to be excluded when selecting species to be transformed by the rule (i.e., empty HRas).

A second molecule of HRas binds the GEF domain of Sos1, where it serves as a 'substrate,' i.e., its binding to GTP/GDP is regulated allosterically by Sos1. Sos1 has different affinities for GTP-loaded and GDP-loaded HRas:

```
HRas(GTPase~GTP) + Sos1(GEF) <->
HRas(GTPase~GTP!1).Sos1(GEF!1)                                      (24a)
k_on_GEFGTP,k_off_GEFGTP
```

```
HRas(GTPase~GDP) + Sos1(GEF) <->
HRas(GTPase~GDP!1).Sos1(GEF!1)                                      (24b)
k_on_GEFGDP,k_off_GEFGDP
```

GDP and GTP interact reversibly with HRas. This process can occur spontaneously (a,b). Sos1 catalyzes exchange of GDP for GTP. Sos1 has lowest activity when not bound to an activator (c), intermediate activity when bound to GDP-HRas (d), and highest activity when bound to GTP-HRas (e). These contextual differences are accounted for in the following rules, with different kinetic parameters accounting for different rates:

```
HRas(GTPase~0) <-> HRas(GTPase~GDP) k_on_GDP,k_off_GDP                   (25a)
```

This journal is © The Royal Society of Chemistry 2011

```
HRas(GTPase~0) <-> HRas(GTPase~GTP) k_on_GTP,k_off_GTP
```
(25b)

```
Sos1(REM,GEF!1).HRas(GTPase~GDP!1) ->
Sos1(REM,GEF) + HRas(GTPase~GTP) k_lo
```
(25c)

```
HRas(GTPase~GDP!1).Sos1(REM!1,GEF!2).HRas(GTPase~GDP!2) ->
HRas(GTPase~GDP!1).Sos1(REM!1,GEF) + HRas(GTPase~GTP) k_med
```
(25d)

```
HRas(GTPase~GTP!1).Sos1(REM!1,GEF!2).HRas(GTPase~GDP!2) ->
HRas(GTPase~GTP!1).Sos1(REM!1,GEF) + HRas(GTPase~GTP) k_hi
```
(25e)

Raf-1 binds GTP-loaded HRas:

```
HRas(GTPase~GTP) + Raf1(RBD) <->
HRas(GTPase~GTP!1).Raf1(RBD!1)
k_on_RasRaf,k_off_RasRaf
```
(26)

This journal is © The Royal Society of Chemistry 2011



Fig. 9A: Alternate subunits

The two subunits can be represented as states of a virtual regulatory subunit.

```
APCC(Reg~Cdh1~Cdc20)
```
(27a)

Alternatively, each subunit can be represented as a separate molecule that can interact with the APC/C core.

```
APCC(core!1).Cdh1(c_box!1), APCC(core!1).Cdc20(c_box!1)
```
(27b)

Fig. 9B: Multiple modifications

Multiple modifications are represented as internal states of the modified component.

```
H3(K9~0~Ac~Me)
```
(28)

Fig. 9C: Homodimer

An EGFR dimer is formed through interactions between the receptor ectodomains.

```
EGFR(ecto!1).EGFR(ecto!1)
```
(29)

Fig. 9D: Overlapping linear motifs

The proline-rich sequence (PRS) and Y188 are represented as separate components.

```
CD3E(PRS,Y188~0~P)
```
(30)

This journal is © The Royal Society of Chemistry 2011

The effect of overlapping motifs is captured in the context of rules. For example, the following rule indicates that Nck binds the PRS of CD3ϵ only when Y188 is not phosphorylated:

```
CD3E(PRS,Y188~0) + Nck(SH3) <->
CD3E(PRS!1,Y188~0).Nck(SH3!1)                    (31)
k_on_CD3E,k_off_CD3E
```

Fig. 9E: Discontinuous binding

Binding sites are represented as virtual components.

```
streptavidin(b,b,b,b)                            (32)
```

References

1. Faeder JR, Blinov ML, Hlavacek WS: **Rule-based modeling of biochemical systems with BioNetGen**. *Methods Mol Biol* 2009, **500**:113–167.
2. Segel LA: **On the validity of the steady state assumption of enzyme kinetics**. *Bull Math Biol* 1988, **50**:579–593.
3. Segel LA, Slemrod M: **The quasi steady-state assumption: a case study in perturbation**. *SIAM Rev* 1989, **31**:446–447.
4. Chen WW, Niepel M, Sorger PK: **Classic and contemporary approaches to modeling biochemical reactions**. *Genes Dev* 2010, **24**:1861–1875.
5. Harris LA, Hogg JS, Faeder JR: **Compartmental rule-based modeling of biochemical systems**. In *Proceedings of the 2009 Winter Simulation Conference*. Edited by Rossetti MD, Hill RR, Johansson B, Dunkin A, Ingalls R 2009:908–919.